Всероссийская олимпиада по информатике 2010/2011 года (Новгородская область)

Решения второго тура муниципального этапа для 9-11 классов
Задача 4. Сжатие последовательности
Очевидно, что разрешенная последовательность (дающая в сумме N) состоит из числа от 1 до A, стоящего на первом месте и разрешенной последовательности, дающей в сумме число от N-1 до N-A, следующей за ним. Пустую последовательность с суммой 0 тоже будем считать разрешенной.

Поэтому, если обозначить функцию, дающую ответ на задачу как R(N), где N – требуемая сумма, то вычислить ее можно по следующей формуле: [image: image2.png]



При ее вычислении «в лоб» будут возникать вызовы функции с одинаковым аргументом, поэтому задачу следует решать с помощью динамического программирования, т.е. запоминать уже найденные значения функции для меньших значений аргумента в массиве.
Особенность входных данных заключается в том, что при заданных ограничениях значение результата не помещается ни в один из стандартных типов данных, поэтому необходимо реализовывать арифметические операции с длинными числами.
Текст решения:

type

  Num=record

   len: integer;

   d: array[1..1000] of byte;

  end;

var

  a, n, i, j: integer;

  r: array[0..1000] of Num;

procedure add(var a:Num; const b:Num);

var s, i, len: integer;

begin

 len:=a.len;

 if len < b.len then len:=b.len;

 for i:=1 to len do begin

  s:=a.d[i] + b.d[i];

  if s > 9 then begin

    a.d[i + 1]:=a.d[i + 1] + 1; 

    s:=s - 10;

  end;

  a.d[i]:=s;

 end;

 if a.d[len + 1] > 0 then len:=len + 1;

 a.len:=len;

end;

begin

 assign(input, 'input.txt');

 reset(input);

 assign(output, 'output.txt');

 rewrite(output);

 readln(n, a);

 r[0].len:=1;

 r[0].d[1]:=1;

 for i:=1 to n do begin

  r[i].len:=0;

  for j:=i-1 downto i-a do

    if j >= 0 then

      add(r[i], r[j]);

 end;

 for i:=r[n].len downto 1 do

   write(r[n].d[i]);

end.

Задача 5. Гирлянда
В этой задаче необходимо перебрать все возможные варианты раскраски лампочек, проверяя каждый из них на допустимость, используя заданный во входных данных список связей между лампочками. На проверку корректности раскраски будет затрачено порядка O(M) операций, количество возможных вариантов раскраски равняется KN (это значение не больше, чем 88=224, что помещается в тип longint). Таким образом, временная сложность алгоритма может быть выражена как O(M∙KN). Перебор всех вариантов может быть выполнен любым удобным способом, например, с помощью рекурсивной процедуры.
Текст решения:

var

 n, k, m, i, r: longint;

 c: array[1..8] of integer;

 w: array[1..10] of record u, v: integer; end;

procedure find(t: integer);

var 

  i: integer;

  ok: boolean;

begin

 if t > n then begin

  ok:=true;

  for i:=1 to m do

   if c[w[i].u] = c[w[i].v] then ok:=false;

  if ok then

   r:=r + 1;

 end else begin

  for i:=1 to k do begin

    c[t]:=i;

    find(t + 1);

  end;

 end;

end;

begin

 assign(input, 'input.txt');

 reset(input);

 assign(output, 'output.txt');

 rewrite(output);

 readln(n, k, m);

 for i:=1 to m do begin

   readln(w[i].u, w[i].v);

 end;

 r:=0;

 find(1);

 writeln(r);

end.

Задача 6. Инопланетные числа
Заданные ограничения входных данных позволяют использовать для хранения чисел встроенные типы (например, longint). Поэтому, достаточно перевесте закодированное число в десятичную систему счисления, а затем сконвертировать его в ту систему, которая задана во входных данных.

Текст решения:

var

  i, j, t, n: integer;

  num: longint;

  c: char;

  r, s, s1, s2: string;

begin

 assign(input, 'input.txt');

 reset(input);

 assign(output, 'output.txt');

 rewrite(output);

 readln(s);

 readln(s1);

 readln(s2);

 num:=0;

 for i:=1 to length(s) do begin

  j:=1;

  while s1[j]<>s[i] do 

   inc(j);

  num:=num * length(s1) + j - 1;

 end;

 r:='';

 while num > 0 do begin

   r:=s2[num mod length(s2) + 1] + r;

   num:=num div length(s2);

 end;

 writeln(r);

end.

